

Enabling SRM Workflow for External Purchase Requisitions



Applies to:

SAP SRM 5.0 (Application controlled workflow), SRM 6.0, SRM 7.0 (Application controlled workflow & Process controlled workflow)

Summary

External requisitions from SAP or Non-SAP systems sent to SRM system are created directly in approved status and will be sent to Sourcing cockpit. This document provides the procedure to enable SRM workflow approval for external requisitions (for approval) of external requisitions before sending these requirements into sourcing cockpit.

Author: Prasanna Rangaraju

Company: SAP Global Delivery, SAP India

Created on: 20 April 2009

Author Bio



Prasanna Rangaraju is working as SRM Consultant in SAP Global Delivery, Bangalore. He has around 6 years of experience. He has undertaken projects for customers based in Sweden, France and US. His experience also includes working as a Solution Management lead in SAP Labs India.


Table of Contents

Enabling SRM Workflow for External Requisitions.....	3
Step 1: Implicit Enhancement in form SSIS_RELEASE_ACTIVITY_GET in Include LBBP_PDFB2.....	5
Step 2: Implicit Enhancement in form ACTIVITY_CHECK_WO_CHECK in Include LBBP_PDF4H.....	6
Step 3: Implicit Enhancement in form PROCDOC_WFL_UPDATE in Include LBBP_PDF2K.....	7
Step 4: Implicit Enhancement in form PROCDOC_WFL_UPDATE_PREPARE in Include LBBP_PDF3M...	8
Step 5: Implement BADI BBP_DOC_SAVE_BADI.....	9
Step 6: Implement BADI BBP_SRC_DETERMINE	10
Test Results.....	11
Step 1: Create a Purchase Requisition in the ERP system.....	11
Step 2: Transfer the Purchase Requisition from ERP to SRM system.....	12
Step 3: Check the Shopping Cart created in SRM system.....	13
Step 4: Approve the Shopping cart by logging onto Portal.....	14
Step 5: Check again the status of Shopping Cart.....	15
Related Content.....	16
Copyright.....	17

Enabling SRM Workflow for External Requisitions

SAP SRM can be used as a central purchasing system that can work together with one/more SAP ERP system and Non-SAP systems and procure any indirect and direct requirements that have been generated in systems other than SAP SRM.

For ex: MRP run in SAP ERP system generates purchase requisitions which can be transferred to SRM system using the report BBP_EXTREQ_TRANSFER. This transfer program creates a shopping cart in the SRM system in "Approved" status and sends the shopping cart directly to Sourcing cockpit. Using sourcing cockpit, Sourcing can be performed and follow-on document can be created (for ex: purchase order).



In the standard SRM system, the external requirements will be directly created in “Approved” status with no Workflow approval. It is assumed that the requisitions are already approved in the source systems. In case the approval of requisitions has to be carried out in SRM instead of source systems, it is currently not possible in the standard SRM system as the requisitions are directly created in approved status.

SRM provides enhanced workflow features and capabilities when compared to other systems and hence it gives more flexibility and value-add if SRM workflows can be reused for External requisitions. Additionally if development was already done for shopping cart workflows, the same could be reused for External requisitions too which will avoid the time to build workflows in source systems. This will be a major advantage if there are multiple source systems which transfer requisitions to a single SRM system.

This document gives the procedure to achieve this by implementing enhancements and BADI implementations. Implicit enhancement concept will be used to achieve this functionality. The following enhancements are applicable for both Process controlled (\geq SRM 6.0) and Application controlled (\leq SRM 5.0).

Based on the start conditions in Application controlled workflow /Process configuration in Process controlled workflow, the corresponding workflow will be started. Once the shopping cart is approved, the shopping cart will be sent to Sourcing cockpit for sourcing. From this point on-ward the functionality is same as that of the standard SRM system.

Step 1: Implicit Enhancement in form SSIS_RELEASE_ACTIVITY_GET in Include LBBP_PDFB2

The form routine SSIS_RELEASE_ACTIVITY_GET is a PD (Procurement Document) Layer form routine which will decide the status of shopping cart. The standard code as shown below directly releases (activity = RELS) the shopping cart in case of external requirements.

```

* -----*
  IF iv_object_type EQ c_shop.

*   For external demands we set the status directly on 'released'
*   If admin change sth. the SC should stay on 'release'
  IF
    iv_subtype EQ c_subtype_er OR
    iv_save_action EQ gc_correct_by_admin.
    ev_next_activity = c_a_sc_release.
  ELSE.
*   if the status is already 'released' we must ask the workflow
*   tool if it will starts a new workflow. If not release directly
  PERFORM check_workflow_restart USING
                                iv_guid
                                iv_method
                                iv_with_change_approval
                                CHANGING
                                lv_restart.

  IF lv_restart EQ c_off.
    ev_next_activity = c_a_sc_release.
  ENDIF.
ENDIF.

```

An implicit enhancement will be created at the start of this form routine where the status is set to TREL (In release) similar to that of normal shopping:

```

ENHANCEMENT 1 ZZ_ENABLE_WORKFLOW_FOR_PR.   "active version
DATA: lv_restart_1          TYPE boolean.

  IF iv_activity EQ c_a_scorder AND iv_bapimode IS NOT INITIAL AND iv_subtype EQ c_subtype_er.
*   First make the process - In Release (Awaiting approval)
    ev_next_activity = c_a_rel_process.

    IF iv_save_action EQ gc_correct_by_admin.
      ev_next_activity = c_a_sc_release.
    ELSE.
*   if the status is already 'released' we must ask the workflow
*   tool if it will starts a new workflow. If not release directly
    PERFORM check_workflow_restart USING
        iv_guid
        iv_method
        iv_with_change_approval
    CHANGING
        lv_restart_1.
    IF lv_restart_1 EQ c_off.
      ev_next_activity = c_a_sc_release.
    ENDIF.
  ENDIF.
*   Return to avoid standard code processing..
  RETURN.
ENDIF.
ENDENHANCEMENT.

```

Step 2: Implicit Enhancement in form ACTIVITY_CHECK_WO_CHECK in Include LBBP_PDF4H

The form routine ACTIVITY_CHECK_WO_CHECK checks whether the status for a particular objects (Shopping cart/Purchase Order etc.,) are valid. For external requirements the status TREL (In Release) is invalid and hence an error message is raised as shown below:

```

WHEN c_a_rel_process.
  IF iv_object_type EQ c_shop.
    CALL FUNCTION 'BBP_HDRADM_GET_HEADER'
      EXPORTING
        iv_guid      = iv_guid
      IMPORTING
        es_header    = lv_header.
    IF lv_header-subtype EQ c_subtype_er.
      * For external demands status is set to 'released' directly
      CLEAR et_messages.
      et_messages-msgty = c_msgty_e.
      et_messages-msgid = 'BBP_PD'.
      et_messages-msgno = 583.
      CALL FUNCTION 'BAL_DSP_TXT_MSG_READ'
        EXPORTING
          i_msgid      = et_messages-msgid
          i_msgno      = et_messages-msgno
        IMPORTING
          e_message_text = et_messages-message.
      APPEND et_messages.
      IF c_on = c_off. MESSAGE e583(bbp_pd) . ENDIF.
      ev_error_activity = c_on.
    ENDIF.
  ENDIF.

```

An implicit enhancement will be created at the end of this form routine where the error message is deleted along with clearing the error activity so that the status TREL becomes a valid status for an external requirement:

```

ENHANCEMENT 3 ZZ_ENABLE_WORKFLOW_FOR_PR.      "active version
IF iv_activity = c_a_rel_process AND iv_object_type = c_shop AND lv_header-subtype EQ c_subtype_er.
* TREL is now valid for External requirements too - so we delete the error
  CLEAR ev_error_activity.
  DELETE et_messages WHERE msgid = 'BBP_PD' AND msgno = 583.
ENDIF.
ENDENHANCEMENT.

```

Step 3: Implicit Enhancement in form PROCDOC_WFL_UPDATE in Include LBBP_PDF2K

This is the form routine where the workflow will be saved to DB in an update task. The standard code executes the workflow save only if the subtype ≠ "ER" (External Requirement) as there is a check as shown below:

```

DATA: lv_user TYPE syuname,
      lv_guid TYPE bbp_guid_32.

CHECK ( iv_object_type EQ c_shop      AND
        iv_subtype IS INITIAL )      OR
        iv_object_type EQ c_po        OR
        iv_object_type EQ c_grse      OR
        iv_object_type EQ c_invoice   OR
        iv_object_type EQ c_contract  OR
        iv_object_type EQ c_quotation OR
        iv_object_type EQ c_pco       OR
        iv_object_type EQ c_biddingdoc.

* no workflow for external demands
CHECK iv_subtype NE c_subtype_er.

```

An implicit enhancement will be created at the start of this form routine where the check for subtype will be removed so that the workflow will be triggered and saved to DB:

```

ENHANCEMENT 2 ZZ_ENABLE_WORKFLOW_FOR_PR.      "active version

DATA: lv_user_l TYPE syuname,
      lv_guid_l TYPE bbp_guid_32.

CHECK ( iv_object_type EQ c_shop ) OR "      AND
"      iv_subtype IS INITIAL )      OR
iv_object_type EQ c_po      OR
iv_object_type EQ c_grse      OR
iv_object_type EQ c_invoice   OR
iv_object_type EQ c_contract  OR
iv_object_type EQ c_quotation OR
iv_object_type EQ c_pco       OR
iv_object_type EQ c_biddingdoc.

** no workflow for external demands
*CHECK iv_subtype NE c_subtype_er.

IF iv_object_type = c_shop.
*Status 11138 (to be reviewed) is no longer needed
*(status is no longer set, but it has been set in "old"
*documents)
*=> delete it
PERFORM status_to_be_reviewed_reset
TABLES it_x_status
USING iv_guid
      iv_object_id.
ENDIF.

*Call workflow interface
IF NOT iv_wfl_action IS INITIAL.
lv_user_l = sy-uname.
MOVE iv_guid TO lv_guid_l.
CALL FUNCTION 'BBP_PDH_WFL_DB_UPDATE'
EXPORTING
  iv_object_id = iv_object_id
  iv_object_type = iv_object_type
  iv_action = iv_wfl_action
  iv_obj_changed = iv_obj_changed
  iv_empl_review = iv_empl_review
  iv_workitem_id = iv_workitem_id
  iv_user = lv_user_l
  iv_guid = lv_guid_l.
ENDIF.

* Return to avoid standard code processing
RETURN.
ENDENHANCEMENT.

```

Step 4: Implicit Enhancement in form PROCDOC_WFL_UPDATE_PREPARE in Include LBBP_PDF3M

The form routine PROCDOC_WFL_UPDATE_PREPARE evaluates which workflow to be launched and prepares for it based on the start conditions or process level configurations. Like in all above form routines, the logic here isn't executed for external requirements:

```

DATA: lv_tot_val_changed TYPE xfeld,
      ls_header           TYPE bbp_pds_header.

CHECK ( cv_object_type EQ c_shop AND
        iv_subtype IS INITIAL ) OR
        cv_object_type EQ c_po OR
        cv_object_type EQ c_grse OR
        cv_object_type EQ c_invoice OR
        cv_object_type EQ c_contract OR
        cv_object_type EQ c_quotation OR
        cv_object_type EQ c_pco OR
        cv_object_type EQ c_biddingdoc.

```

An implicit enhancement at the start of this form routine is created where this check will be removed. The code excerpt below gives the commented logic that will be part of the enhancement.

```

| ENHANCEMENT 4 ZZ_ENABLE_WORKFLOW_FOR_PR.      "active version
DATA: lv_tot_val_changed_l TYPE xfeld,
      ls_header_l         TYPE bbp_pds_header.

CHECK cv_object_type EQ c_shop OR "AND
      "iv_subtype IS INITIAL ) OR
      cv_object_type EQ c_po OR
      cv_object_type EQ c_grse OR
      cv_object_type EQ c_invoice OR
      cv_object_type EQ c_contract OR
      cv_object_type EQ c_quotation OR
      cv_object_type EQ c_pco OR
      cv_object_type EQ c_biddingdoc.

*Determine header information
| IF it_x_hdradm IS INITIAL.
  CALL FUNCTION 'BBP_HDRADM_GETDETAIL'
  EXPORTING
    i_guid = iv_header_guid
  CHANGING
    e_header = ls_header_l.

```

The following procedure can be followed to create this enhancement:

1. Create an enhancement at the start of the form routine
2. Copy the code in the form routine into enhancement
3. Remove the check on subtype for shopping cart as shown above
4. Add a RETURN statement before the end of ENHANCEMENT to avoid the execution of standard logic

Step 5: Implement BADI BBP_DOC_SAVE_BADI

In the standard SRM processing, the external requirement is always sent to sourcing cockpit as the indicator SOURCE_REL_IND at item level will be set to "X."

When the workflow is activated for external requirements with above enhancements, it is required to make sure that the shopping cart is sent to sourcing cockpit only after the shopping cart is approved. The BADI BBP_DOC_SAVE_BADI will be implemented to achieve this. As shown below, in this BADI implementation the shopping cart status will be checked. If shopping cart is not yet approved, clear the SOURCE_REL_IND and if approved set the SOURCE_REL_IND.

This will make sure that the external requirement becomes available in sourcing cockpit only when it is approved.

```

METHOD if_ex_bbp_doc_save_badi~bbp_doc_save.

  DATA : lt_item          TYPE TABLE OF bbp_pds_sc_item_d,
         ls_item          TYPE bbp_pds_sc_item_d,
         lt_status        TYPE TABLE OF bbp_pds_status,
         lv_source_rel_ind TYPE bbp_pdigp-source_rel_ind,
         ls_header        TYPE bbp_pds_sc_header_d.

  CALL FUNCTION 'BBP_PD_SC_GETDETAIL'
    EXPORTING
      i_guid = iv_doc_guid
    IMPORTING
      e_header = ls_header
    TABLES
      e_item = lt_item
      e_status = lt_status.

  * Check if shopping cart is an external requirement
  CHECK ls_header-subtype = 'ER'.

  READ TABLE lt_status TRANSPORTING NO FIELDS WITH KEY stat = 'I1129' inact = space.
  IF sy-subrc = 0.
  * Shopping cart (External Requirement) approved
    lv_source_rel_ind = 'X'.
  ELSE.
  * Shopping cart (External Requirement) is not yet approved
    CLEAR lv_source_rel_ind.
  ENDIF.

  * Loop on items and set the sourcing indicator accordingly..
  LOOP AT lt_item INTO ls_item.
    CALL FUNCTION 'BBP_PDH_DB_DIRECT_UPDATE'
      EXPORTING
        iv_item_guid_sourcing = ls_item-guid
        iv_source_rel_ind      = lv_source_rel_ind.
    ENDLOOP.

ENDMETHOD.

```

Step 6: Implement BADI BBP_SRC_DETERMINE

As mentioned above in the standard SRM system, the external requirements are always created directly in approved status and sent to sourcing cockpit. But with above enhancements workflow is triggered so that an approver has to approve the external requirement before it becomes available in sourcing cockpit.

During the approval procedure, once the shopping cart is completely approved the transfer module tries to create a follow-on document based on configuration. This might create a Purchase order/Purchase requisition in local or backend system.

In order to avoid creation of a follow-on document directly for an external requirement, the sourcing flag will be set (based on subtype) so that it will be sent to sourcing cockpit without any automatic follow-on document creation.

This is achieved by implementing the BBP_SC_DETERMINE BADI with the logic as shown below:

```
METHOD if_ex_bbp_src_determine~determine_sourcing.  
  
  DATA : ls_header TYPE bbp_pds_sc_header_d.  
  
  CALL FUNCTION 'BBP_PD_SC_GETDETAIL'  
    EXPORTING  
      i_guid    = header_bbp_guid  
    IMPORTING  
      e_header = ls_header.  
  
  IF ls_header-subtype = 'ER'.  
    sourcing = 'X'.  
  ENDIF.  
  
ENDMETHOD.
```

Test Results

The following test case describes the target behavior in an SRM 7.0 system after implementing all the above enhancements. The same behavior is applicable even for other versions of SRM system.

Step 1: Create a Purchase Requisition in the ERP system

Create a purchase requisition with a Material Group/Purchasing Group which is enabled for external transfer (configured in views V_T160PR, V_T160EX) as shown below:



The screenshot shows the SAP 'Create Purchase Requisition' interface. At the top, there is a title bar 'Create Purchase Requisition' and a menu bar with options like 'Document Overview On' and 'Personal Setting'. Below the menu bar, there are input fields for 'NB Purch.requis. Stand' and a checkbox for 'Source Determination'. A 'Header' section contains a toolbar with various icons and a 'Default Values' button. The main area is a table with the following columns: St..., Item A, I, Material, Short Text, Quantity, Unit, C, Delivery Date, Matl Group, Plant, Stor. Loc., PGr, Requisnr., Tracking..., Des.Vendor, and Fixed Venc. The first row of data contains the following values: Item A: 10, I: K, Material: External Requisition Approval, Quantity: 10, Unit: EA, C: D, Delivery Date: 19.10.2009, Matl Group: ESA legal, Plant: Atlanta, Stor. Loc.: 001.


St...	Item A	I	Material	Short Text	Quantity	Unit	C	Delivery Date	Matl Group	Plant	Stor. Loc.	PGr	Requisnr.	Tracking...	Des.Vendor	Fixed Venc
	10	K		External Requisition Approval	10	EA	D	19.10.2009	ESA legal	Atlanta		001				

Step 2: Transfer the Purchase Requisition from ERP to SRM system

Transfer the created purchase requisition to SRM system using the report BBP_EXTREQ_TRANSFER:

Program BBP_EXTREQ_TRANSFER

Selection purch. requisition	10013498	to		
Transfer date	20.04.2009			
Offset Transfer Date in Days				
Packet size for selection	100			
Max. No. of Purch. Req. Items	100000			
Name of Queue (QRFC)	BBP_EXTREQ_TRANS			
<input type="checkbox"/> Several Queues				

The result of transfer is shown below:

Program BBP_EXTREQ_TRANSFER	
Program BBP_EXTREQ_TRANSFER	
Number of transferred purchase requisitions:	1
Number of transferred purchase requisition items:	1

Step 3: Check the Shopping Cart created in SRM system

Check the created shopping cart in SRM system to verify that the shopping cart is awaiting approval with workflow triggered. Note that the Shopping cart is not yet available in Sourcing cockpit as it is not yet approved.

```

Header:
GUID                DE2D902A8E4F25F1AA71001A64C6B06A
Object_ID           469
Object_Type         BUS2121      EC Requirement Coverage Request
Process_Type        EXTR         External Requirement
Template_Type
Version_Type
Posting_Date        20.04.2009
Description         TRA RFC 20.04.2009 11:46
Logical_System      SRM_00_300
Created_At          20.04.2009 11:46:53 by TRA RFC
Changed_At          20.04.2009 11:46:53 by TRA RFC
-> Details

Items:
Pos  Type  Quantity  Unit  Ordered_Prod  Description  Gross price  Price
0000000001  10      EA        External Requisition Approval  11,00  11,00

Stats:
Status  Description  Inactiv
HEADER  I1015        Awaiting Approval
HEADER  I1021        Created
HEADER  I1038        Complete
HEADER  I1106        Shopping cart ordered

Partners:
Partner_Fct  Description  Partner_No  Excluded
0000000001  00000016    Requester   0017FEB7DDA4AB488B64EB28890D248E
0000000001  00000020    Goods Recipient  0017FEB7DDA4AB488B64EB28890D248E
0000000001  00000027    Delivery Point  11C8C8B1EC92C046917370422C35C8A7
0000000001  00000075    Location     BF944FF791BDAC45B1AF7590B69768DD

Organizations:
Proc_Org  Proc_Group  Proc_Org_Resp
0000000001  0 50001766  0 50000614  0 50000614

Header Links:
Obj_Key  Obj_Type  LogSys  RoleType  Obj_Key  Obj_Type  LogSys  RoleType
Obj_Key  Obj_Type  LogSys  RoleType  Obj_Key  Obj_Type  LogSys  RoleType
HEADER  BUS2121  SRM_00_300  BBP_TARGET <<<< 0010013498  0000000000  EXTREQ  T90CLNT090  BBP_SOURCE

Item Links:
Obj_Key  Obj_Type  LogSys  RoleType  Obj_Key  Obj_Type  LogSys  RoleType
Obj_Key  Obj_Type  LogSys  RoleType  Obj_Key  Obj_Type  LogSys  RoleType
0000000001  BUS2121001  SRM_00_300  BBP_TRITM <<<< 0010013498  00000000100000000000  EXTREQ01  T90CLNT090  BBP_SRITM

Contracts:
Ctr_Hdr_Guid  Ctr_Hdr_Number  Ctr_Guid  Ctr_Item_Number
    
```

The workflow will be triggered based on the Start condition or Process level configuration:

```

Workflow Item:
Workitem-ID        00000078745
Task               WS40000014
Text              SRM Approval Process for Document ID 469 (GUID: DE2D902A8E4F25F1AA71001A64C6B06A)
Status            STARTED
Statustext
Date              20.04.2009
Time              11:46:53
No Workflow Item Found
    
```

Step 4: Approve the Shopping cart by logging onto Portal

Now login to Portal as approver, open the work item from Universal Work List (UWL) and approve the shopping cart:

Subject	!	From	Sent	Priority	Due	Status
Approve Shopping Cart 469 with Value 110.00 USD		TRA_RFC,	Today	Normal	3	New
Error in MW-BDoc-Processing, Type PRODUCT_MAT ID 3AB13D71229B5D428E5378954BB2E5B5		Workflow-System,	Mar 15, 2005	Normal	1	New

Approve Shopping Cart:469

Number 469 **Document Name** TRA_RFC 20.04.2009 11:46 **Status** Awaiting Approval **Created On** 20.04.2009 11:46:53

[Approve](#) [Reject](#) [Close](#) [Save](#) [Check](#) [System Info](#) [Create Memory Snapshot](#)

General Data
 Buy on Behalf Of: Approval Note:
 Name of Shopping Cart:
 Default Settings: [Set Values](#)
 Approval Process: [Display / Edit Agents](#) Note to Supplier:
 Document Changes: [Display](#)

Item Overview
[Details](#) [Process All Items](#)

Line Number	Item Type	Product ID	Description	Product Category	Product Category Description	Quantity
0001	Material		External Requisition Approval	00704	ESA legal	10

Step 5: Check again the status of Shopping Cart

Check the status of shopping cart again to verify that the shopping cart is now approved and is available in Sourcing cockpit:

```

Header:
GUID DE2D902A8E4F25F1AA71001A64C6B06A
Object_ID 469
Object_Type BUS2121 EC Requirement Coverage Request
Process_Type EXTR External Requirement
Template_Type
Version_Type Active Document
Posting_Date 20.04.2009
Description TRA_RFC 20.04.2009 11:46
Logical_System SRM_00_300
Created_At 20.04.2009 11:46:53 by TRA_RFC
Changed_at 20.04.2009 11:50:55 by EMPLOYEE-00
-> Details

Items:
Pos Type Quantity Unit Ordered_Prod Description Gross price Price
0000000001 10 EA External Requisition Approval 11,00 11,00

Stats:
Status Description Inactiv
HEADER I1015 Awaiting Approval X
HEADER I1021 Created
HEADER I1038 Complete
HEADER I1106 Shopping cart ordered
HEADER I1129 Approved

Partner:
Partner_Fct Description Partner_No Excluded
0000000001 00000016 Requester 0017FEB7DDA4AB488B64EB28890D248E
0000000001 00000020 Goods Recipient 0017FEB7DDA4AB488B64EB28890D248E
0000000001 00000027 Delivery Point 11C8C8B1EC92C046917370422C35C8A7
0000000001 00000075 Location BF944FF791BDAC45B1AF7590B69768DD

Organizations:
Proc_Org Proc_Group Proc_Org_Resp
0000000001 0 50001766 0 50000614 0 50000614

Header Links:
Obj_Key Obj_Type LogSys Roletype Obj_Key Obj_Type LogSys Roletype
HEADER BUS2121 SRM_00_300 BBP_TRITM <<<< 0010013498 0000000000 EXTREQ T90CLNT090 BBP_SOURCE

Item Links:
Obj_Key Obj_Type LogSys Roletype Obj_Key Obj_Type LogSys Roletype
0000000001 BUS2121001 SRM_00_300 BBP_TRITM <<<< 0010013498 0000000001000000000000 EXTREQ01 T90CLNT090 BBP_SRITM
0000000001 Item Displayed in Sourcing Cockpit
    
```

Related Content

[Enhancement Concept](#)

[Creating Source code Plug-In using implicit enhancement](#)

[Procurement of External Requirements](#)

Copyright

© Copyright 2009 SAP AG. All rights reserved.

No part of this publication may be reproduced or transmitted in any form or for any purpose without the express permission of SAP AG. The information contained herein may be changed without prior notice.

Some software products marketed by SAP AG and its distributors contain proprietary software components of other software vendors.

Microsoft, Windows, Excel, Outlook, and PowerPoint are registered trademarks of Microsoft Corporation.

IBM, DB2, DB2 Universal Database, System i, System i5, System p, System p5, System x, System z, System z10, System z9, z10, z9, iSeries, pSeries, xSeries, zSeries, eServer, z/VM, z/OS, i5/OS, S/390, OS/390, OS/400, AS/400, S/390 Parallel Enterprise Server, PowerVM, Power Architecture, POWER6+, POWER6, POWER5+, POWER5, POWER, OpenPower, PowerPC, BatchPipes, BladeCenter, System Storage, GPFS, HACMP, RETAIN, DB2 Connect, RACF, Redbooks, OS/2, Parallel Sysplex, MVS/ESA, AIX, Intelligent Miner, WebSphere, Netfinity, Tivoli and Informix are trademarks or registered trademarks of IBM Corporation.

Linux is the registered trademark of Linus Torvalds in the U.S. and other countries.

Adobe, the Adobe logo, Acrobat, PostScript, and Reader are either trademarks or registered trademarks of Adobe Systems Incorporated in the United States and/or other countries.

Oracle is a registered trademark of Oracle Corporation.

UNIX, X/Open, OSF/1, and Motif are registered trademarks of the Open Group.

Citrix, ICA, Program Neighborhood, MetaFrame, WinFrame, VideoFrame, and MultiWin are trademarks or registered trademarks of Citrix Systems, Inc.

HTML, XML, XHTML and W3C are trademarks or registered trademarks of W3C®, World Wide Web Consortium, Massachusetts Institute of Technology.

Java is a registered trademark of Sun Microsystems, Inc.

JavaScript is a registered trademark of Sun Microsystems, Inc., used under license for technology invented and implemented by Netscape.

SAP, R/3, SAP NetWeaver, Duet, PartnerEdge, ByDesign, SAP Business ByDesign, and other SAP products and services mentioned herein as well as their respective logos are trademarks or registered trademarks of SAP AG in Germany and other countries.

Business Objects and the Business Objects logo, BusinessObjects, Crystal Reports, Crystal Decisions, Web Intelligence, Xcelsius, and other Business Objects products and services mentioned herein as well as their respective logos are trademarks or registered trademarks of Business Objects S.A. in the United States and in other countries. Business Objects is an SAP company.

All other product and service names mentioned are the trademarks of their respective companies. Data contained in this document serves informational purposes only. National product specifications may vary.

These materials are subject to change without notice. These materials are provided by SAP AG and its affiliated companies ("SAP Group") for informational purposes only, without representation or warranty of any kind, and SAP Group shall not be liable for errors or omissions with respect to the materials. The only warranties for SAP Group products and services are those that are set forth in the express warranty statements accompanying such products and services, if any. Nothing herein should be construed as constituting an additional warranty.